
Chapter 7 

 

 
 7.2: Shift and Rotate Instructions 

Q Which instruction shifts each bit in an operand to the left and copies the highest bit into 

both the Carry flag and the lowest bit position? 

 

A ROL. 

 

Q Which instruction shifts each bit to the right, copies the lowest bit into the Carry flag, and 

copies the Carry flag into the highest bit position? 

 

A RCR. 

 

Q Write a sequence of shift instructions that cause AX to be sign-extended into EAX. In 

other words, the sign bit of AX is copied into the upper 16 bits of EAX. 

 

A First, shift EAX 16 bits to the left. Next, shift it arithmetically 16 bits to the right: 

shl eax,16 

sar eax,16 

 

Q Which instruction performs the following operation (CF = Carry flag)? 

Before: CF, AL = 1  11010101 

After:   CF, AL = 1  10101011 

 

A RCL. 

 

Q Suppose the instruction set contained no rotate instructions. Show how we might use 

SHR and a conditional jump instruction to rotate the contents of the AL register one 

position to the right. 

 

A      shr al,1       ; shift AL into Carry flag 

     jnc next       ; Carry flag set? 

     or al,80h      ; yes: set highest bit 

next:               ; no: do nothing 

Integer Arithmetic - Part1 

Q What happens to the Carry flag when the SHR AX,1 instruction is executed? 

 

A The Carry flag receives the lowest bit of AX (before the shift).  

 

Q Write a logical shift instruction that multiplies the contents of EAX by 16. 

 

A shl eax,4 

 

Q Write a logical shift instruction that divides EBX by 4. 

 

A shr ebx,2 

 

Q Write a single rotate instruction that exchanges the high and low halves of the DL register. 

 

A ror dl,4       ;(or: rol dl,4) 

 

Q Write a SHLD instruction that shifts the highest bit of the AX register into the lowest bit 

position of DX and shifts DX one bit to the left. 

 

A shld dx,ax,1 

 

Q In the following code sequence, show the value of AL after each shift or rotate instruction 

has  executed: 

mov al,0D4h 

shr al,1            ; a. 

mov al,0D4h 

sar al,1            ; b. 

mov al,0D4h 

sar al,4            ; c. 

mov al,0D4h 

rol al,1            ; d. 

  

A (a) 6Ah  (b) 0EAh  (c)  0FDh  (d) A9h 

 

Q In the following code sequence, show the value of AL after each shift or rotate instruction 

has  executed: 

mov al,0D4h 

ror al,3      ; a. 

mov al,0D4h 

rol al,7      ; b. 

stc 

mov al,0D4h 

rcl al,1      ; c. 

stc 

mov al,0D4h 

rcr al,3      ; d. 

A (a) 9Ah  (b) 6Ah  (c) 0A9h  (d) 3Ah 

Q Write a series of instructions that shift the lowest bit of AX into the highest bit of BX 

without using the SHRD instruction. Next, perform the same operation using SHRD. 

 

A shr ax,1      ; shift AX into Carry flag 

rcr bx,1      ; shift Carry flag into BX 
; Using SHRD: 
shrd bx,ax,1 

Q One way to calculate the parity of a 32-bit number in EAX is to use a loop that shifts each 

bit into the Carry flag and accumulates a count of the number of times the Carry flag was 

set. Write a code that does this, and set the Parity flag accordingly. 

 

A     mov ecx,32       ; loop counter 

    mov bl,0         ; counts the '1' bits 
L1: shr eax,1        ; shift into Carry flag 
    jnc L2           ; Carry flag set? 
    inc bl           ; yes: add to bit count 
L2: loop L1          ; continue loop 
    ; if BL is odd, clear the parity flag 
    ; if BL is even, set the parity flag 
    shr bl,1 
    jc odd 
    mov bh,0 
    or bh,0          ; PF = 1 
    jmp next 
odd: 

mov bh,1 
or bh,1         ; PF = 0 

next: 

 

 7.3: Shift and Rotate Applications 

Q Write a sequence of instructions that shift three memory bytes to the right by 1 bit 

position. Use the following data definition: 

          byteArray BYTE 81h,20h,33h 

 

A This problem requires us to start with the high-order byte and work our way down to the 

lowest byte: 

byteArray BYTE 81h,20h,33h 

.code 

shr byteArray+2,1 

rcr byteArray+1,1 

rcr byteArray,1 

 

 

Q Write a sequence of instructions that shift three memory words to the left by 1 bit position. 

Use the following data definition: 

wordArray WORD 810Dh, 0C064h,93ABh 

 

A This problem requires us to start with the low-order word and work our way up to the  

    highest word: 

wordArray WORD 810Dh,0C064h,93ABh 

.code 

shl wordArray,1 

rcl wordArray+2,1 

rcl wordArray+4,1 

Q Write ASM instructions that calculate EAX * 24 using binary multiplication. 

 

A The multiplier (24) can be factored into 16 + 8 = (24+23): 
mov ebx,eax    ; save a copy of eax 

shl eax,4      ; multiply by 16 

shl ebx,3      ; multiply by 8 

add eax,ebx    ; add the products 

 

Q Write ASM instructions that calculate EAX * 21 using binary multiplication. 

 

A The multiplier (21) can be factored into 16 + 4 + 1 = (24+22+20): 
mov ebx,eax     ; save a copy of eax 

mov ecx,eax     ; save another copy of eax 

shl eax,4       ; multiply by 16 

shl ebx,2       ; multiply by 4 

add eax,ebx     ; add the products 

add eax,ecx     ; add original value of eax 

 

 Displaying Binary Bits 

Algorithm: Shift MSB into the Carry flag; If CF = 1, append a "1" character to a string; 

otherwise, append a "0" character. Repeat in a loop, 32 times. 

 

.data 

buffer db 32 DUP(0) 

.code 

   mov cx,32 

   mov si,OFFSET buffer 

L1: shl eax,1 

   mov BYTE PTR [si],'0' 

   jnc L2 

   mov BYTE PTR [si],'1' 

L2: inc si 

   loop L1 

 

Q What change would you make to the previous algorithm in order to display the binary bits 

in reverse order? 

 

A Change the instruction at label L1 to shr eax,1 

 

 

 

 

 Isolating MS-DOS File Date Fields 

The MS-DOS file date field packs the year, month, and day into 16 bits, If a file was last 

modified on March 10, 1999, the file’s date stamp would appear as follows in the DX register 

(the year number is relative to 1980): 

 

 

 

 

 

 

o Isolate the Day field: 

mov al,dl           ; make a copy of DL 

and al,00011111b    ; clear bits 5-7 

mov day,al          ; save in day 

 

o Isolate the Month field: 

mov ax,dx           ; make a copy of DX 

shr ax,5            ; shift right 5 bits 

and al,00001111b    ; clear bits 4-7 

mov month,al        ; save in month 

 

o Isolate the Year field: 

mov al,dh           ; make a copy of DH 

shr al,1            ; shift right one position 

mov ah,0            ; clear AH to zeros 

add ax,1980         ; year is relative to 1980 

mov year,ax         ; save in year 

 

Q The time stamp field of a file directory entry uses bits 0 through 4 for the seconds, bits 5 

through 10 for the minutes, and bits 11 through 15 for the hours. Write instructions that 

extract the minutes and copy the value to a byte variable named bMinutes. 

 

A We will assume that the time stamp word is in the DX register: 

mov ax,dx 

shr ax,5 

and al,00111111b    ; (leading zeros optional) 

mov bMinutes,al     ; save in variable 


